Tema 9
Cálculo integral de funciones reales de variable real

Objetivos:

1. Calcular funciones primitivas con wxMaxima.
2. Practicar con el concepto de función integrable y la integral de una función.
3. Trabajar con funciones definidas mediante integrales
4. Calcular integrales de forma aproximada.
5. Estudiar algunas aplicaciones de la integral.
6. Estudiar la generalización de la integral (integrales impropias).

Contenidos:

09-1. Antiderivadas o primitivas de una función. Integral indefinida.
09-2. Integral de una función en un intervalo. Cálculo de integrales.
09-3. La función integral. Aplicaciones
09-4. Cálculo aproximado de integrales. Integración numérica.
09-5. Integrales impropias.

Referencias

AEM11 ALANINOS PRATS, J; EXTREMERA LIZANA, J; MUÑOZ RIVAS, P. (2011)
Cálculo con wxMaxima.

APJ11 ALANINOS PRATS, J; APARICIO DEL PRADO, C; EXTREMERA LIZANA, J; MUÑOZ RIVAS, P.; VILLENA MUÑOZ, A.R. (2011)
Prácticas de ordenador con wxMaxima.

AP86 APOSTOL, T.M. (1986)
Análisis Matemático

BR09 BRUZÓN GALLEGO, M. DE LOS SANTOS; RAMÍREZ LABRADOR, JOSÉ (2009)
Modelos matemáticos con Maxima

BU07 DE BURGOS, JUAN (2007)
Cálculo Infinitesimal de una variable (segunda edición).

ES08 ESTELA CARBONELL, M. ROSA; SAÀ SEOANE, JOEL (2008)
Cálculo con soporte interactivo en moodle.

ES02 ESTEP, DONALD (2002)
Practical Analysis in one variable

GV07a GONZÁLEZ VEGA, LAUREANO (2007)
Laboratorio de Matemáticas. Vol. 1: números y ecuaciones

GV07b GONZÁLEZ VEGA, LAUREANO (2007)
Laboratorio de Matemáticas. Vol. 2: límites y derivadas

JB01 JARAUTA BRAGULAT, EUSEBI (2001)
Anàlisi Matemàtica d'una variable. Fonaments i aplicacions.

RR05 REDONDO NEBLE, M. VICTORIA; RODRÍGUEZ GALVÁN, J. RAFAEL (2005)
Introducción a Maxima

RR08b RODRÍGUEZ RIOTORTO, MARIO (2008)
Curso intensivo i-MATH de software libre orientado a Ciencias e Ingeniería

RU80 RUDIN, WALTER (1980)
Principios de Análisis Matemático.

SP95 SPIVAK, MICHAEL (1995)
Calculus (Càlcul Infinitesimal).

VR09 VALLEJO RODRÍGUEZ, JOSÉ ANTONIO (2009)
Cálculo diferencial con Maxima

09-1.- Antiderivadas o primitivas de una función. Integral indefinida.

Los contenidos de este apartado se desarrollan en el archivo **Tema_09-1.wxm**.

Definición. Si $f:[a,b]\subset\mathbb{R}\to\mathbb{R}$ es una función acotada en un intervalo $[a,b]$, se llama <u>antiderivada o primitiva</u> de f a cualquier función F continua en $[a,b]$, derivable en $]a,b[$ y tal que $DF(x)=f(x)$ para todo $x\in]a,b[$. Así por ejemplo, $F_1(x)=x^3+2$ i $F_2(x)=x^3-6$ son antiderivadas de la función $f(x)=3x^2$ en cualquier intervalo $[a,b]\subset\mathbb{R}$.

Si F es una antiderivada de *f*, es inmediato razonar que el conjunto de todas las antiderivadas de f se llama <u>integral indefinida</u> de f y se define mediante:

$$D^{-1}(f(x))=\{F(x)+C,\ DF(x)=f(x), C=\text{constant}\}$$

También es habitual designar la integral indefinida de una función con la notación clásica $\int f(x)dx$, notación que se justificará más adelante.

La metodología clásica para la obtención de la integral indefinida de una función se fundamenta en dos resultados importantes: la fórmula de integración por partes y el teorema del cambio de variable, que enunciamos a continuación.

Cálculos de primitivas por partes. Si se trata de calcular la integral indefinida de una función *f*(*x*), se puede escribir esta función como el producto $f(x)=u(x)Dv(x)$ y entonces se puede calcular la integral indefinida de f mediante la expresión:

$$D^{-1}\left(f(x)\right)=D^{-1}\left(u(x)Dv(x)\right)=u(x)v(x)-D^{-1}\left(Du(x)v(x)\right)+C$$

La notación clásica de este método es:

$$\int f(x)dx=\int u(x)Dv(x)dx=u(x)v(x)-\int Du(x)v(x)dx+C$$

Cálculos de primitivas por cambio de variable. Para calcular la integral indefinida de una función continua $f:[a,b]\subset\mathbb{R}\to\mathbb{R}$ se considera una función $\varphi:[c,d]\to[a,b]$ biyectiva y derivable en el intervalo $]c,d[$ tal que $D\varphi(t)\neq 0$ para todo $t\in]c,d[$. Esta función se llama función de cambio de variable. Por lo tanto, si $x=\varphi(t)$ esto equivale a $t=\varphi^{-1}(x)$ y además la función inversa φ^{-1} es derivable en $]a,b[$, verificándose que

$$D\varphi^{-1}(x)=\frac{1}{D\varphi(t)}$$

Si se considera ahora una antiderivada G de la función $g(t)=f(\varphi(t))D\varphi(t)$ se cumple que:

$$D\left(G\circ\varphi^{-1}\right)(x)=DG\left(\varphi^{-1}(x)\right)\cdot D\varphi^{-1}(x)=f\left(\varphi(t)\right)D\varphi(t)\frac{1}{D\varphi(t)}=f(x)$$

Así pues, el cálculo de una antiderivada de f se puede llevar a cabo calculando una antiderivada de la función $g(t)$ y entonces se cumple:

$$D^{-1}\left(f(x)\right)=\left(D^{-1}\left(g(t)\right)\right)\circ\varphi^{-1}(x)+C$$

expresión que se conoce como antiderivación o primitivación por <u>cambio de variable o por substitución</u>.

La notación clásica de este método es:

$$\int f(x)dx=\left(\int f(g(t))Dg(t)dt\right)\circ\varphi^{-1}(x)+C$$

Vamos a ver cómo se desarrolla el cálculo de antiderivadas o primitivas con wxMaxima. En primer lugar hay que definir la función a integrar y después la instrucción que hay que aplicar es “integrate” especificando la función y la variable. Así, por ejemplo:

```
(%i1) f1(x):=2*x^2-5*x+1;     integrate(f1(x), x);
```

(%o1) $f1(x):=2\,x^2-5\,x+1$

(%o2) $\frac{2\,x^3}{3}-\frac{5\,x^2}{2}+x$

El resultado es la primitiva que podríamos llamar “canónica”, ya que la constante de integración es nula. Si se quiere una notación en la que se vea este proceso de forma más clara, entonces:

```
(%i3) 'integrate(f1(x), x)=integrate(f1(x), x);
```

(%o3) $\int 2\,x^2-5\,x+1\,dx=\frac{2\,x^3}{3}-\frac{5\,x^2}{2}+x$

También se puede calcular la integral indefinida de funciones en la expresión de las cuales hay parámetros. Por ejemplo:

```
(%i4) f2(x) := a2*x^2 + a1*x + a0;
                    ' integrate(f2(x), x)= integrate(f2(x), x);
```

(%o4) $f2(x):=a2\,x^2+a1\,x+a0$

(%o5) $$\int a2\,x^2+a1\,x+a0\,dx\ =\frac{a2\,x^3}{3}+\frac{a1\,x^2}{2}+a0\,x$$

Veamos otro ejemplo:

```
(%i6) assume(a>0)$
      f3(x):=x^a;        ' integrate(f3(x), x)=integrate(f3(x), x);
```

(%o7) $f3(x):=x^a$

(%o8) $$\int x^a\,dx\ =\frac{x^{a+1}}{a+1}$$

```
(%i9) f33(x):=x^(-1);        ' integrate(f33(x), x)=integrate(f33(x), x);
```

(%o9) $f33(x):=x^{-1}$

(%o10) $$\int \frac{1}{x}dx\ =\log(x)$$

En ocasiones el programa no da la respuesta que se podría esperar; así, por ejemplo:

```
(%i11) f4(x):=log(1+sqrt(1+x^2))$
       'integrate(f4(x), x)=integrate(f4(x), x);
```

(%o12) $$\int \log(\sqrt{x^2+1}+1)\,dx\ =\int \frac{x^2}{(x^2+1)^{3/2}+x^2+1}dx\ +x\log(\sqrt{x^2+1}+1)+\mathrm{atan}(x)-x$$

Hay pues que evaluar de alguna manera la integral indefinida que aparece en el segundo miembro. Veamos su cálculo; en primer lugar definimos la función a integrar:

```
(%i13) R(x):=x^2/((x^2+1)^(3/2)+x^2+1)$        integrate(R(x) ,x);
```

(%o14) $$\int \frac{x^2}{(x^2+1)^{3/2}+x^2+1}dx$$

Ahora calculamos una expresión de esta función aplicando una simplificación algebraica de esta función racional:

```
(%i15) r1:x^2/((x^2+1)^(3/2)+x^2+1)$        r1=ratsimp(r1),algebraic;
```

(%o16) $$\frac{x^2}{(x^2+1)^{3/2}+x^2+1}=\frac{\sqrt{x^2+1}-1}{x^2+1}$$

Y ahora ya podemos expresarla como función y calcular su integral indefinida:

```
(%i17) r2(x):=(sqrt(x^2+1)-1)/(x^2+1)$
       'integrate(r2(x), x)=integrate(r2(x), x);
```

$$\text{(\%o18)} \quad \int \frac{\sqrt{x^2+1}-1}{x^2+1}\,dx = \operatorname{asinh}(x)-\operatorname{atan}(x)$$

Finalmente ya podemos expresar la integral indefinida que se quería calcular:

```
(%i19) 'integrate(f4(x), x)=asinh(x)-atan(x)+x*log(sqrt(x^2+1)+1)+atan(x)-x;
```

$$\text{(\%o19)} \quad \int \log\left(\sqrt{x^2+1}+1\right)dx = x\log\left(\sqrt{x^2+1}+1\right)+\operatorname{asinh}(x)-x$$

Verificamos la corrección de los cálculos, viendo que en efecto, la derivada de la función primitiva es la función inicial:

```
(%i20) F4(x):=x*log(sqrt(x^2+1)+1)+asinh(x)-x$
       diff(F4(x), x);
```

$$\text{(\%o21)} \quad \log\left(\sqrt{x^2+1}+1\right)+\frac{x^2}{\sqrt{x^2+1}\left(\sqrt{x^2+1}+1\right)}+\frac{1}{\sqrt{x^2+1}}-1$$

```
(%i22) radcan(%);
```

$$\text{(\%o22)} \quad \log\left(\sqrt{x^2+1}+1\right)$$

Veamos algunos ejemplos más:

```
(%i23) f5(x):=1/((x-2)*sqrt(x+2))$
       'integrate(f5(x), x)=integrate(f5(x), x);
```

$$\text{(\%o24)} \quad \int \frac{1}{(x-2)\sqrt{x+2}}\,dx = 2\left(\frac{\log\left(\sqrt{x+2}-2\right)}{4}-\frac{\log\left(\sqrt{x+2}+2\right)}{4}\right)$$

```
(%i25) f6(x):=(2*x+a)/(x^2-5*x+3)$
       'integrate(f6(x), x)=integrate(f6(x), x);
```

$$\text{(\%o26)} \quad \int \frac{2x+a}{x^2-5x+3}\,dx = \frac{(a+5)\log\left(\frac{2x-\sqrt{13}-5}{2x+\sqrt{13}-5}\right)}{\sqrt{13}}+\log\left(x^2-5x+3\right)$$

```
(%i27) f7(x):=(b*x+a)/(x^2-5*x+3)$
       'integrate(f7(x), x)=integrate(f7(x), x);
```

$$\text{(\%o28)} \quad \int \frac{bx+a}{x^2-5x+3}\,dx = \frac{(5b+2a)\log\left(\frac{2x-\sqrt{13}-5}{2x+\sqrt{13}-5}\right)}{2\sqrt{13}}+\frac{b\log\left(x^2-5x+3\right)}{2}$$

```
(%i29) ratsimp(%);
```

$$(\%o29)\quad \int \frac{b\,x+a}{x^2-5\,x+3}\,dx = \frac{(5\,b+2\,a)\,\log\left(\frac{2\,x-\sqrt{13}-5}{2\,x+\sqrt{13}-5}\right)+\sqrt{13}\,b\,\log(x^2-5\,x+3)}{2\,\sqrt{13}}$$

En algunos casos el cálculo de la integral indefinida no se puede llevar a cabo, ya que pese a que exista no se puede expresar en términos de funciones elementales. Veamos un ejemplo clásico al respecto:

```
(%i30) f8(x):=exp(-x^2)$
       'integrate(f8(x), x)=integrate(f8(x), x);
```

$$(\%o31)\quad \int \%e^{-x^2}\,dx = \frac{\sqrt{\pi}\,\mathrm{erf}(x)}{2}$$

09-2.- Integral de una función en un intervalo. Cálculo de integrales.

Los contenidos de este apartado se desarrollan en el archivo **Tema_09-2.wxm**.

Si $f: I=[a,b]\subset\mathbb{R}\to\mathbb{R}$ es una función positiva y continua y, por lo tanto, acotada en el intervalo cerrado $[a,b]$, la integral definida, o sencillamente la integral, de la función en el intervalo se calcula mediante el Teorema fundamental del Cálculo o regla de Barrow, el cual establece que:

$$\int_{[a,b]} f(x)dx \equiv \int_a^b f(x)dx = F(b)-F(a) = \left[F(x)\right]_a^b$$

donde F es una antiderivada o primitiva de f en $[a,b]$.

Si $f: I=[a,b]\subset\mathbb{R}\to\mathbb{R}$ es una función positiva y continua a trozos en el intervalo cerrado $[a,b]$, con discontinuidades evitables o de salto en este intervalo, es decir, si $I_k=[x_k,x_{k+1}],\ k=1,2,\dots,p$, $I_1\cup\cdots\cup I_p=I$, y los límites laterales de la función en los puntos frontera de estos intervalos son finitos, si indicamos por $f(x)=f_k(x),\ x\in I_k,\ k=1,2,\dots,p$, i si $D^{-1}\left(f_k(x)\right)=F_k(x),\ x\in I_k$, entonces:

$$\int_{[a,b]} f(x)dx = \sum_{k=1}^{p} \int_{[x_k,x_{k+1}]} f_k(x)dx$$

$$\int_{[x_k,x_{k+1}]} f_k(x)dx = F_k(x_{k+1})-F_k(x_k),\ \ k=1,2,\dots,p$$

La sintaxis para el cálculo de la integral de una función en un intervalo cerrado con wxMaxima es muy sencilla y consiste en definir la función y aplicar la instrucción "integrate" indicando la función, la variable y los límites inferior y superior del intervalo de integración:

```
(%i1) f1(x):=3*x^2 - 2*x + 4$
      'integrate(f1(x), x, 1, 3)=integrate(f1(x), x, 1, 3);
```

(%o2) $\int_1^3 3x^2-2x+4\,dx = 26$

```
(%i3) f2(x):=sin(omega*x)$
      'integrate(f2(x), x, a, b)=integrate(f2(x), x, a, b);
```

(%o4) $\int_a^b \sin(\omega x)\,dx = \frac{\cos(a\omega)}{\omega}-\frac{\cos(b\omega)}{\omega}$

```
(%i5) f3(x):=exp(alpha*x)$
      'integrate(f3(x), x, a, b)=integrate(f3(x), x, a, b);
```

(%o6) $$\int_a^b \%e^{\alpha x} dx = \frac{\%e^{\alpha b}}{\alpha} - \frac{\%e^{a \alpha}}{\alpha}$$

Sin embargo, no siempre se obtiene el resultado que se espera:

```
(%i7) f4(x):=cos(x)/(100 - x^6)$
      'integrate(f4(x), x, 0, 1)=integrate(f4(x), x, 0, 1);
```

(%o8) $$\int_0^1 \frac{\cos(x)}{100 - x^6} dx = \int_0^1 \frac{\cos(x)}{100 - x^6} dx$$

Es decir, el programa no siempre calcula la integral de cualquier función y, por lo tanto, hay que desarrollar otras metodologías que permitan este cálculo. La mejor es la que se denomina integración numérica o cálculo aproximado de integrales, que se verá más adelante. Aquí podemos obtener una primera aproximación aplicando el Teorema del valor medio integral, el cual establece que

$$\int_a^b f(x)dx = \mu(b-a), \quad m = \min_{[a,b]} f(x) \le \mu \le M = \max_{[a,b]} f(x)$$

Se define el valor medio integral mediante:

$$\mu = \frac{1}{(b-a)} \int_a^b f(x)dx$$

En el caso anterior nos ayudamos de una representación gráfica:

```
(%i3) wxplot2d([f(x)], [x,0, 1], [y, 0, 0.015], [ylabel, ""] , [legend, "f(x)"] , [style, [lines, 2]] ,
      [gnuplot_preamble, "set size ratio 1; set grid"] ,  [nticks, 300]    )$
```

(%t3)



Se observa que la función es decreciente y por lo tanto, su máximo se alcanza en el extremo inferior del intervalo y el mínimo en el extremo superior del intervalo:

```
(%i4) M=M:f(0);          m=m:f(1);
      f(0), numer;       f(1), numer;
```

(%o4) $M=\frac{1}{100}$

(%o5) $m=\frac{\cos(1)}{99}$

(%o6) 0.01

(%o7) 0.0054575990491731

Tomando como valor intermedio la media aritmética del máximo y del mínimo, resulta:

```
(%i8) Integral_aproximada=IA:(m+M)/2;
```

(%o8) $Integral\_aproximada=\frac{\frac{\cos(1)}{99}+\frac{1}{100}}{2}$

```
(%i9) %, numer;
```

(%o9) Integral_aproximada=0.0077287995245866

La representación gráfica es:

```
(%i10) IA(x):=0.0077287995245866$
       wxplot2d([f(x), IA(x)], [x,0, 1], [y, 0, 0.015], [legend, "f(x)" , "IA(x)"], [ylabel, ""] , [style, [lines, 2]] ,
       [gnuplot_preamble, "set size ratio 1; set grid"] , [nticks, 300] )$
```

(%t11)



En el caso de una función periódica sinusoidal, se llama valor eficaz de la función la raíz cuadrada del valor medio integral del cuadrado de la función, es decir:

$$V_{ef}=\sqrt{\frac{1}{(b-a)}\int_a^b (f(x))^2\,dx}$$

Ilustrémoslo gráficamente con un ejemplo:

```
(%i12) g(x):=(sin(x))^2$
       'integrate(g(x), x, 0, 2*%pi)=integrate(g(x), x, 0, 2*%pi);
```

(%o13) $\int_0^{2\pi} \sin(x)^2 dx = \pi$

```
(%i14) Vef=Vef:sqrt((1/(2*%pi))*integrate(g(x), x, 0, 2*%pi));
```

(%o14) $Vef = \frac{1}{\sqrt{2}}$

```
(%i15) wxplot2d([g(x), Vef], [x, 0, 2*%pi], [y, 0, 1.5] , [legend, "g(x)" , "Vef"], [ylabel, ""] , [style, [lines, 2]] ,
   [gnuplot_preamble, "set size ratio 1; set grid"] , [nticks, 300]    )$
```

(%t15) 

09-3.- La función integral. Aplicaciones.

Los contenidos de este apartado se desarrollan en el archivo **Tema_09-3.wxm**.

Si $f: I=[a,b]\subset\mathbb{R}\to\mathbb{R}$ es una función integrable en un intervalo cerrado y acotado, se define la <u>función integral o función área</u> mediante:

$$F(x)=\begin{cases}\int_a^x f(t)dt, & \text{si } a<x\le b\\ 0, & \text{si } x=a\end{cases}$$

Esta función tiene dos propiedades muy importantes:

- Es continua en el intervalo $[a,b]$.
- Si la función f es continua en el intervalo $[a,b]$, entonces la función integral F es derivable en $]a,b[$ verificándose que $DF(x)=f(x),\ \forall x\in\,]a,b[$.

Con wxMaxima se puede definir la función integral como se expone a continuación. En primer lugar definir la función, utilizando una variable diferente y a continuación calcular la integral definida poniendo como límite inferior el extremo inferior del intervalo y como extremo superior la variable que se asignará a la función integral. Veamos la sintaxis:

```
(%i1) F(x):=integrate(f(t), t, a, x);
```

(%o1) $F(x):=\int_a^x f(t)dt$

Ejemplo 9.3.1. Consideramos la función $f: I=[-3,3]\subset\mathbb{R}\to\mathbb{R}$ definida por:

$$f(t)=2t^2-3t+4,\ t\in[-3,3]$$

En primer lugar lo expresamos mediante una integral:

```
(%i1) f(t):=2*t^2 - 3*t + 4$
      '( integrate(f(t), t, -3, x) ) = integrate(f(t), t, -3, x) ;
```

(%o2) $\int_{-3}^x f(t)dt=\frac{4x^3-9x^2+24x}{6}+\frac{87}{2}$

Ahora definimos la función integral:

(%i3) define (F(x) , integrate(f(t), t, -3, x)) ;

(%o3) $F(x):=\frac{4x^3-9x^2+24x}{6}+\frac{87}{2}$

Una representación gráfica conjunta con las dos funciones puede ayudar a la interpretación; obsérvese que para hacer la representación gráfica hay que redefinir la función f usando la misma variable que para la función integral. Así:

```
(%i3) f(x):=2*x^2 - 3*x + 4$
      wxplot2d([f(x),F(x)], [x, -3, 3], [y, 0, 70] , [legend, "f(x)" , "F(x)"] , [style, [lines, 2]] ,
       [plot_format, gnuplot],  [gnuplot_preamble, "set size ratio 1; set grid"] , [nticks, 300] )$
```

(%t4)



Dado que la función f es continua, la función integral F es derivable y se ha de cumplir la propiedad enunciada anteriormente. En efecto, calculamos esta derivada:

(%i5) diff(F(x), x);

(%o5) $\frac{12x^2-18x+24}{6}$

(%i6) ratsimp(%);

(%o6) $2x^2-3x+4$

Obsérvese que el valor de la función integral en el extremo superior del intervalo ha de coincidir con la integral de la función en el intervalo. En efecto:

(%i7) 'integrate(f(t), t, -3, 3)=integrate(f(t), t, -3, 3);
'F(3)=F(3);

(%o7) $\int_{-3}^{3} 2t^2-3t+4\,dt=60$

(%o8) $F(3)=60$

Ejemplo 9.3.2. Consideramos la función $f: I = [0,4] \subset \mathbb{R} \to \mathbb{R}$ definida por:

$$f(x) = \begin{cases} 6, & \text{si } 0 \le x < 2 \\ 3, & \text{si } 2 \le x \le 4 \end{cases}$$

Esta función no es continua y observaremos que la función integral es continua pero no derivable en el intervalo de definición. En efecto:

```
(%i1) f1(t):=6$        f2(t):=3$
      f(t):=if t<=2 then f1(t) else f2(t);
(%o3) f(t):=if  t<=2  then  f1(t)  else  f2(t)
(%i4) integrate(f1(t), t, 0, x);
      integrate(f2(t), t, 2, x);
(%o4) 6 x
(%o5) 3 (x-2)

(%i6) f1(x):=6$        f2(x):=3$
      f(x):=if x<=2 then f1(t) else f2(t)$
(%i9) F1(x):=6*x$      F2(x):=F1(2)+3*(x-2)$
      F(x):=if x<=2 then F1(x) else F2(x);
(%o11) F(x):=if  x<=2  then  F1(x)  else  F2(x)
(%i12) wxplot2d([f(x), F(x)], [x, 0, 4], [y, 0, 25] , [legend, "f(x)" , "F(x)"] , [style, [lines, 2]] ,
       [plot_format, gnuplot],  [gnuplot_preamble, "set size ratio 1; set grid"] , [nticks, 300])$
```

(%t12)



Aplicaciones de la función integral.

La definición de la función integral permite calcular derivadas de funciones definidas mediante integrales, es decir, funciones de la forma:

$$G(x)=\int_a^{u(x)} f(t)dt \ ; \ H(x)=\int_{v(x)}^{u(x)} f(t)dt$$

En efecto, aplicando la regla de la cadena es inmediato establecer que la derivada de estas funciones es:

$$DG(x)=D\left(\int_a^{u(x)} f(t)dt\right)=f\left(u(x)\right)Du(x)$$

$$DH(x)=D\left(\int_{v(x)}^{u(x)} f(t)dt\right)=f\left(u(x)\right)Du(x)-f\left(v(x)\right)Dv(x)$$

Veamos como se hacen los cálculos con wxMaxima.

Ejemplo 9.3.3. Se trata de calcular la derivada de la función definida por:

$$G(x)=\int_2^{e^x} \frac{\log(1+t)}{2t^2}dt$$

Entonces:

```
(%i1) f(t):=(log(1+t))/(2*t^2);
      F(x):=integrate(f(t), t, 2, %e^x);
```

$$\text{(\%o1) } f(t):=\frac{\log(1+t)}{2\,t^2}$$

$$\text{(\%o2) } F(x):=\int_2^{\%e^x} f(t)dt$$

```
(%i3) assume(%e^x - 2 > 0)$
      DF(x)=diff(F(x), x);
```

$$\text{(\%o4) } DF(x)=\frac{\%e^{-x}\left(\left(\%e^x+1\right)\log\left(\%e^x+1\right)-x\,\%e^x\right)-\%e^{-x}\left(\%e^x\log\left(\%e^x+1\right)-x\,\%e^x\right)}{2}$$

```
(%i5) ratsimp(%);
```

$$\text{(\%o5) } DF(x)=\frac{\%e^{-x}\log\left(\%e^x+1\right)}{2}$$

Ejemplo 9.3.4. Se trata de calcular la derivada de la función definida por:

$$H(x)=\int_{x+1}^{3x^2+2x-1}\left(1+e^t\right)dt$$

Entonces:

```
(%i1) F3(x):=integrate(f(t), t, u(x), v(x));
```

(%o1) $F3(x) := \int_{u(x)}^{v(x)} f(t)dt$

```
(%i2) f(t):=exp(t)+1$    u(x):=x+1$      v(x):=3*x^2+2*x-1$
      diff(F3(x),x);
```

(%o5) $\%e^{-1}\left((6x+2)\%e^{3x^2+2x}+6\%e\,x+2\%e\right)-\%e^{x+1}-1$

```
(%i6) ratsimp(%);
```

(%o6) $\%e^{-1}\left((6x+2)\%e^{3x^2+2x}-\%e^{x+2}+6\%e\,x+\%e\right)$

09-4.- Cálculo aproximado de integrales. Integración numérica.

Los contenidos de este apartado se desarrollan en el archivo **Tema_09-4.wxm**.

Si $f:[a,b]\subset\mathbb{R}\to\mathbb{R}$ es integrable en este intervalo, el Teorema fundamental del Cálculo permite calcular la integral de la función en $[a,b]\subset\mathbb{R}$. Sin embargo, hay funciones para las que el cálculo de su integral indefinida es muy difícil o incluso no es expresable mediante funciones elementales, como es por ejemplo el caso de la función $\exp(-x^2)$ que es continua en cualquier intervalo $[a,b]\subset\mathbb{R}$ y, por lo tanto, admite una primitiva. Para poder resolver el cálculo de la integral en estos casos, se pueden aplicar métodos aproximados, como los que se estudian en este apartado.

Si $f:[a,b]\subset\mathbb{R}\to\mathbb{R}$ es una función integrable en $[a,b]\subset\mathbb{R}$, se denomina <u>partición</u> del intervalo un conjunto de puntos $P=\{x_0,x_1,...,x_n\}$ del intervalo $[a,b]$ tal que $h=x_{k+1}-x_k, k=0,1,...,n-1$. Naturalmente se cumple $h=\dfrac{b-a}{n}$.

9.4.1. El método de los trapecios.

En cada uno de los intervalos $I_k=[x_k,x_{k+1}],\ k=0,1,...,n-1$ asociados a la partición se considera la función g_k definida por la recta que pasa por los puntos $(x_k,f(x_k))$ y $(x_{k+1},f(x_{k+1}))$; Hecho esto, la integral de la función dada en el intervalo I_k se puede aproximar por la integral de esta función lineal en I_k. Finalmente, la integral de la función en $[a,b]\subset\mathbb{R}$ se puede aproximar por la suma de las integrales de g_k en I_k :

$$\int_a^b f(x)dx \approx \sum_{k=0}^{n-1}\int_{x_k}^{x_{k+1}} g_k(x)dx = h\left(\frac{f(a)+f(b)}{2}+\sum_{k=1}^{n-1} f(a+kh)\right)$$

expresión que se conoce como *fórmula de los trapecios*.

Ejemplo 9.4.1. Se quiere calcular aproximadamente por la fórmula de los trapecios la integral de la función $f(x)=1+\dfrac{\cos(x)}{100-x^6}$ en el intervalo $[0,2]$.

Haremos los cálculos afinando sucesivamente la partición del intervalo viendo en primer lugar que el programa no nos da ningún resultado si pedimos calcular la integral:

(%i1) f(x):=1 + cos(x)/(100-x^6)$
'integrate(f(x), x, 0, 2)=integrate(f(x), x, 0, 2);

(%o2) $$\int_0^2 \frac{\cos(x)}{100-x^6}+1\,dx = \int_0^2 \frac{\cos(x)}{100-x^6}+1\,dx$$

Haremos una representación gráfica de la función en el intervalo considerado:

(%i3) wxplot2d([f(x)], [x, 0, 2], [y, 0.96, 1.04] , [ylabel, ""] , [legend, "f(x)"] , [style, [lines, 2]] , [plot_format, gnuplot], [gnuplot_preamble, "set size ratio 1; set grid"])$

(%t3) 

Y ahora ya podemos comenzar las iteraciones:

(%i4) h=h:0.2$ n1=2/h;

(%o5) n1 = 10.0

(%i6) IA_trapezis_1=T1:h*((f(0) + f(2))/2 + sum(f(h*n), n, 1, 9)), numer;

(%o6) IA_trapezis_1 = 2.008142250402611

(%i7) kill(h)$ h=h:0.1$ n2=2/h;

(%o9) n2 = 20.0

(%i10) IA_trapezis_2=T2:h*((f(0) + f(2))/2 + sum(f(h*n), n, 1, 19)), numer;

(%o10) IA_trapezis_2 = 2.008332190715854

(%i11) kill(h)$ h=h:0.05$ n3=2/h;

(%o13) n3 = 40.0

(%i14) IA_trapezis_3=T3:h*((f(0) + f(2))/2 + sum(f(h*n), n, 1, 39)), numer;

(%o14) IA_trapezis_3 = 2.008384205938344

```
(%i15) kill(h)$     h=h:0.01$     n4=2/h;
(%o17) n4 = 200.0
(%i18) IA_trapezis_4=T4:h*((f(0) + f(2))/2 + sum(f(h*n), n, 1, 199)), numer;
(%o18) IA_trapezis_4 = 2.008401415164511

(%i19) kill(h)$     h=h:0.005$     n5=2/h;
(%o21) n5 = 400.0
(%i22) IA_trapezis_5=T5:h*((f(0) + f(2))/2 + sum(f(h*n), n, 1, 399)), numer;
(%o22) IA_trapezis_5 = 2.008401958078575
```

9.4.2. El método de los rectángulos.

En cada uno de los intervalos $I_k = [x_k, x_{k+1}],\ k = 0,1,...,n-1$ asociados a la partición se considera la función constante g_k definida por el valor de la función en el punto medio de este intervalo, es decir:

$$g_k(x_k^*) = f\left(\frac{x_k + x_{k+1}}{2}\right), \quad k = 0,1,2,..,n-1$$

Hecho esto, la integral de la función dada en el intervalo I_k se puede aproximar por la integral de esta función constante en I_k. Finalmente, la integral de la función en $[a,b] \subset \mathbb{R}$ se puede aproximar por la suma de las integrales de g_k en I_k :

$$\int_a^b f(x)dx \approx \sum_{k=0}^{n-1} \int_{x_k}^{x_{k+1}} g_k(x)dx = hf(x_k^*)$$

expresión que se conoce como *fórmula de los rectángulos*.

Ejemplo 9.4.2. Se quiere calcular aproximadamente por la fórmula de los rectángulos la integral de la función $f(x) = 1 + \dfrac{\cos(x)}{100 - x^6}$ en el intervalo $[0,2]$.

ya sabemos que el programa no calcula esta integral y tenemos la representación gráfica hecha en el Ejemplo anterior. Ya podemos comenzar directamente las iteraciones:

```
(%i23) kill(h)$     h=h:0.2$     n1=2/h;
(%o25) n1 = 10.0
(%i26) IA_rectangles_1=R1:h*(sum(f(h*(n + n+1)/2), n, 0, 9)), numer;
(%o26) IA_rectangles_1 = 2.008522131029097
```

```
(%i27) kill(h)$        h=h:0.1$        n2=2/h;
(%o29) n2 = 20.0
(%i30) IA_rectangles_2=R2:h*(sum(f(h*(n + n+1)/2), n, 0, 19)), numer;
(%o30) IA_rectangles_2 = 2.008436221160835

(%i31) kill(h)$        h=h:0.05$        n3=2/h;
(%o33) n3 = 40.0
(%i34) IA_rectangles_3=R3:h*(sum(f(h*(n + n+1)/2), n, 0, 39)), numer;
(%o34) IA_rectangles_3 = 2.008411041591628

(%i35) kill(h)$        h=h:0.01$        n4=2/h;
(%o37) n4 = 200.0
(%i38) IA_rectangles_4=R4:h*(sum(f(h*(n + n+1)/2), n, 0, 199)), numer;
(%o38) IA_rectangles_4 = 2.00840250099264
```


```
(%i35) kill(h)$        h=h:0.01$        n4=2/h;
(%o37) n4 = 200.0
(%i38) IA_rectangles_4=R4:h*(sum(f(h*(n + n+1)/2), n, 0, 199)), numer;
(%o38) IA_rectangles_4 = 2.00840250099264
```


9.4.3. El método de Simpson.

Este método se basa en la aproximación de la función por un polinomio de segundo grado. Para poder llevar a cabo esta aproximación se requiere que el número de intervalos sea par y, por lo tanto, que el número de puntos de la partición sea impar. Por lo tanto, indicaremos $n = 2p$, $h = \dfrac{b-a}{2p}$ y los puntos de la partición serán $P = \left\{x_0, x_1, \ldots, x_{2p}\right\}$. La expresión que resulta es:

$$\int_a^b f(x)dx \approx \frac{h}{3}\sum_{k=0}^{p-1}\left(f(x_{2k}) + 4f(x_{2k+1}) + f(x_{2k+2})\right)$$

expresión que se conoce como *fórmula de Simpson*.

Ejemplo 9.4.3. Se quiere calcular aproximadamente por la fórmula de los rectángulos la integral de la función $f(x) = 1 + \dfrac{\cos(x)}{100 - x^6}$ en el intervalo $[0, 2]$.

Ya sabemos que el programa no calcula esta integral y tenemos la representación gráfica hecha antes. Ya podemos comenzar directamente las iteraciones:

```
(%i43) kill(h)$        h=h:0.2$        n1=n1:2/h;     p1=n1/2;
(%o45) n1 = 10.0
  (%o46) p1 = 5.0
(%i47) IA_Simpson_1=
        S1:(h/3)*(sum(f(h*2*n) + 4*f(h*(2*n+1)) + f(h*(2*n+2)), n, 0, 4)), numer;
(%o47) IA_Simpson_1 = 2.008349561013199

(%i48) kill(h)$        h=h:0.1$        n2=n2:2/h;     p2=n2/2;
(%o50) n2 = 20.0
  (%o51) p2 = 10.0
(%i52) IA_Simpson_2=
        S2:(h/3)*(sum(f(h*2*n) + 4*f(h*(2*n+1)) + f(h*(2*n+2)), n, 0, 9)), numer;
(%o52) IA_Simpson_2 = 2.008395504153601

(%i53) kill(h)$        h=h:0.02$        n3=n3:2/h;     p3=n3/2;
(%o55) n3 = 100.0
  (%o56) p3 = 50.0
(%i57) IA_Simpson_3=
        S3:(h/3)*(sum(f(h*2*n) + 4*f(h*(2*n+1)) + f(h*(2*n+2)), n, 0, 49)), numer;
(%o57) IA_Simpson_3 = 2.008402121260181

(%i58) kill(h)$        h=h:0.01$        n4=n4:2/h;     p4=n4/2;
(%o60) n4 = 200.0
  (%o61) p4 = 100.0
(%i62) IA_Simpson_4=
        S4:(h/3)*(sum(f(h*2*n) + 4*f(h*(2*n+1)) + f(h*(2*n+2)), n, 0, 99)), numer;
(%o62) IA_Simpson_4 = 2.008402137973978
```

9.4.4. Métodos implementados en wxMaxima.

El programa wxMaxima tiene implementado el llamado método de Romberg, que es un método iterativo basado en la fórmula de los trapecios (ver referencias). La sintaxis es muy sencilla:

```
(%i63) 'integrate(f(x), x, 0, 2)=romberg(f(x), x, 0, 2);
```

(%o63) $\int_0^2 \frac{\cos(x)}{100-x^6}+1\,dx = 2.008395056340455$

09-5.- Integrales impropias.

Los contenidos de este apartado se desarrollan en el archivo **Tema_09-5.wxm**.

Las integrales impropias son una generalización de la integral de una función en un intervalo, en los casos siguientes:
1) El intervalo no es acotado y la función es acotada en este intervalo (integrales impropias de primera especie).
2) El intervalo es cerrado y la función no es acotada en este intervalo (integrales impropias de segunda especie).

Pueden darse conjuntamente las dos situaciones anteriores, aplicándose entonces una metodología que contempla los dos casos.

9.5.1 Integrales impropias de primera especie.

Consideramos una función $f:[a,+\infty[\,\to \mathbb{R}$ que cumple:

- La función es acotada en el su dominio;
- La función es integrable en todo intervalo $[a,x]\subset[a,+\infty[$ cualquiera que sea $x>a$.

En estas condiciones se define la integral (impropia) de la función en el intervalo $[a,+\infty[$ mediante el límite al infinito de su función integral, es decir:

$$\int_a^{+\infty} f(t)dt = \lim_{x\to+\infty}\int_a^x f(t)dt$$

Si el límite existe y es finito se dice que la integral (impropia) es convergente. En caso contrario, se dice que es divergente.

Se definen de forma análoga:

$$\int_{-\infty}^{b} f(t)dt = \lim_{x\to-\infty}\int_x^b f(t)dt$$

$$\int_{-\infty}^{+\infty} f(t)dt = \lim_{x\to+\infty}\int_c^x f(t)dt + \lim_{x\to-\infty}\int_x^c f(t)dt$$

Ejemplo 9.5.1. Se considera la función $f(x)=\exp(-ax),\ a>0$ definida en el intervalo $[0,+\infty[$. Se trata de ilustrar el método de estudio de la convergencia de esta integral impropia y, si es convergente, calcularla.

En primer lugar definimos la función:

```
(%i1) f1(t):=exp(-a*t);
(%o1) f1(t):= exp((-a) t)
```

Ahora calculamos la función integral:

```
(%i2) assume(a>0)$
      'integrate(f1(t), t, 0, x)=integrate(f1(t), t, 0, x);
```

$$(\%o3)\quad \int_0^x \%e^{-a\,t}dt = \frac{1}{a}-\frac{\%e^{-a\,x}}{a}$$

```
(%i4) F1(x):=1/a-%e^(-a*x)/a;
```

$$(\%o4)\ F1(x):=\frac{1}{a}-\frac{\%e^{(-a)\,x}}{a}$$

Ahora calcularemos el límite de la función integral:

```
(%i5) 'limit(F1(x), x, inf)=limit(F1(x), x, inf);
```

$$(\%o5)\quad \lim_{x\to\infty} \frac{1}{a}-\frac{\%e^{-a\,x}}{a}=\frac{1}{a}$$

Dado que es finito, la integral impropia es convergente y vale 1/*a*. La sintaxis de wxMaxima que permite los cálculos directamente es:

```
(%i6) 'integrate(f1(t), t, 0, inf)=integrate(f1(t), t, 0, inf);
```

$$(\%o6)\quad \int_0^{\infty} \%e^{-a\,t}dt = \frac{1}{a}$$

Ejemplo 9.5.2. Se considera la función $f(x)=\dfrac{1}{x^a}$, $a\geq 1$ definida en el intervalo $[1,+\infty[$. Se trata de ilustrar el método de estudio de la convergencia de esta integral impropia y, si es convergente, calcularla.

En primer lugar definimos la función:

```
(%i1) f2(t):=1/t^a;
```

$$(\%o1)\ f2(t):=\frac{1}{t^a}$$

Dividiremos el estudio en dos casos: $a > 1$ y $a = 1$. Ahora calculamos la función integral en el primer caso:

```
(%i2) assume(a>1)$      assume(x>1)$
      'integrate(f2(t), t, 1, x)=integrate(f2(t), t, 1, x);
```

$$\text{(\%o4)}\quad \int_1^x \frac{1}{t^a}dt = \frac{1}{a-1}-\frac{x^{1-a}}{a-1}$$

```
(%i5) F2(x):=1/(a-1)-x^(1-a)/(a-1);
```

$$\text{(\%o5)}\quad F2(x):=\frac{1}{a-1}-\frac{x^{1-a}}{a-1}$$

Ahora calcularemos el límite de la función integral:

```
(%i6) 'limit(F2(x), x, inf)=limit(F2(x), x, inf);
```

$$\text{(\%o6)}\quad \lim_{x\to\infty}\frac{1}{a-1}-\frac{x^{1-a}}{a-1}=\frac{1}{a-1}$$

Dado que es finito, la integral impropia es convergente y vale 1/*a*-1. La sintaxis de wxMaxima que permite los cálculos directamente es:

```
(%i7) 'integrate(f2(t), t, 1, inf)=integrate(f2(t), t, 1, inf);
```

$$\text{(\%o7)}\quad \int_1^\infty \frac{1}{t^a}dt = \frac{1}{a-1}$$

Veamos ahora el caso a = 1:

```
(%i8) f21(t):=1/t$
      'integrate(f21(t), t, 1, inf)=integrate(f21(t), t, 1, inf);
defint: integral is divergent.
 -- an error. To debug this try: debugmode(true);
```

En este caso la integral impropia es divergente.

9.5.2 Integrales impropias de segunda especie.

Consideramos una función $f:]a,b]\rightarrow\mathbb{R}$ tal que:

- es integrable en todo intervalo cerrado $[a+\delta,b]\subset]a,b]$ cualquiera que sea $0<\delta<b-a$;
- se cumple $\lim_{x\to a+} f(x)=+\infty$.

Entonces, se define la integral impropia de *f* en el intervalo $[a,b]$ mediante:

$$\int_a^b f(t)dt = \lim_{\delta \to 0} \int_{a+\delta}^b f(t)dt$$

Si este límite existe, la integral impropia se dice que es *convergente*; en caso contrario, se dice que es *divergente*.

Se define de forma análoga la integral impropia de la función f en el intervalo $[a,b]$, en los casos:

- La función es integrable en todo intervalo cerrado $[a,b-\delta] \subset [a,b[$ cualquiera que sea $0 < \delta < b-a$ y $\lim_{b-} f = \infty$
- La función es integrable en los intervalos $[a,x_0-\delta]$ y $[x_0+\delta,b]$ cualquiera que sea $0 < \delta < \max\{x_0-a, b-x_0\}$ y $\lim_{x_0-} f = \lim_{x_0+} f = \infty$.

Ejemplo 9.5.3. Se trata de estudiar la convergencia de la integral impropia

$$\int_1^2 \frac{1}{\sqrt{x-1}} dx$$

En primer lugar definimos la función:

(%i1) f4(t):=1/(sqrt(t-1));

(%o1) $f4(t):=\frac{1}{\sqrt{t-1}}$

Ahora calcularemos la función integral:

(%i2) assume(x-2<0)$ assume(x-1>0)$
'integrate(f4(t), t, x, 2)=integrate(f4(t), t, x, 2);

(%o4) $\int_x^2 \frac{1}{\sqrt{t-1}}dt = 2-2\sqrt{x-1}$

(%i5) F4(x):=2-2*sqrt(x-1);

(%o5) $F4(x):=2-2\sqrt{x-1}$

Y finalmente calcularemos el límite lateral por la derecha en el extremo inferior del intervalo:

(%i6) 'limit(F4(x), x, 1, plus)=limit(F4(x), x, 1, plus);

(%o6) $\lim_{x \to 1+} 2-2\sqrt{x-1} = 2$

Por lo tanto, la integral impropia es convergente. La sintaxis para calcularla con wxMaxima es:

```
(%i7) 'integrate(f4(t), t, 1, 2)=integrate(f4(t), t, 1, 2);
```

(%o7) $\int_{1}^{2} \frac{1}{\sqrt{t-1}} dt = 2$

Observación. Consideramos una función $f :]a,b] \to \mathbb{R}$ tal que f es integrable en todo intervalo cerrado $[a+\delta, b] \subset]a,b]$ cualquiera que sea $0 < \delta < b-a$ y se cumple $\lim_{x \to a+} f(x) = +\infty$. Si se considera el cambio de variable definido por:

$$x = g(t) = a + \frac{1}{t}; \quad Dg(t) = -\frac{1}{t^2}$$

entonces se cumple:

$$\int_a^b f(x)dx = \int_{+\infty}^{\frac{1}{b-a}} f\left(a+\frac{1}{t}\right)\left(-\frac{1}{t^2}\right)dt = \int_{\frac{1}{b-a}}^{+\infty} f\left(a+\frac{1}{t}\right)\left(\frac{1}{t^2}\right)dt$$

es decir, la integral impropia de segunda especie se ha transformado en una integral impropia de primera especie.